@PACKAGE_INIT@

# Any changes in this ".cmake" file will be overwritten by CMake, the source is the ".cmake.in" file.

include(CMakeFindDependencyMacro)
find_dependency(KF6CoreAddons "@KF_DEP_VERSION@")
find_dependency(KF6Config "@KF_DEP_VERSION@")

if (NOT @KIOCORE_ONLY@)
    find_dependency(KF6Bookmarks "@KF_DEP_VERSION@")
    find_dependency(KF6Completion "@KF_DEP_VERSION@")
    find_dependency(KF6ItemViews "@KF_DEP_VERSION@")
    find_dependency(KF6JobWidgets "@KF_DEP_VERSION@")
    find_dependency(KF6Solid "@KF_DEP_VERSION@")
    find_dependency(KF6WidgetsAddons "@KF_DEP_VERSION@")
    find_dependency(KF6WindowSystem "@KF_DEP_VERSION@")
    find_dependency(KF6Service "@KF_DEP_VERSION@")

    if (NOT @BUILD_SHARED_LIBS@)
        find_dependency(KF6ConfigWidgets "@KF_DEP_VERSION@")
        find_dependency(KF6GuiAddons "@KF_DEP_VERSION@")
        find_dependency(KF6IconThemes "@KF_DEP_VERSION@")

        if (@HAVE_QTDBUS@)
            find_dependency(KF6DBusAddons "@KF_DEP_VERSION@")
        endif()

        if (NOT ANDROID)
            find_dependency(KF6Crash "@KF_DEP_VERSION@")
        endif()

        if (@LibMount_FOUND@)
            find_dependency(LibMount)
        endif()
    endif()
else()
    if (NOT @BUILD_SHARED_LIBS@)
        find_dependency(KF6Service "@KF_DEP_VERSION@")
    endif()
endif()

find_dependency(Qt6Network "@REQUIRED_QT_VERSION@")
if (@HAVE_QTDBUS@)
    find_dependency(Qt6DBus "@REQUIRED_QT_VERSION@")
endif()

include("${CMAKE_CURRENT_LIST_DIR}/KF6KIOTargets.cmake")
@PACKAGE_INCLUDE_QCHTARGETS@
